Lab Problem 8.1(a), (b), (c) Physics 430

(a)

```
> restart;
```

Here is the proposed solution

```
> T:=To/sqrt(1+4*D*t/sigma^2)*exp(-(x-L/2)^2/(sigma^2+4*D*t));
```

$$T := \frac{To\, \mathbf{e}^{\left(-\frac{(x-1/2\,L)^2}{\sigma^2+4\,D\,t}\right)}}{\sqrt{1+\frac{4\,D\,t}{\sigma^2}}}$$

Calculate the time derivative on the left side of the diffusion equation

```
> left:=diff(T,t);
```

$$left := -2\,\frac{To\, \mathbf{e}^{\left(-\frac{(x-1/2\,L)^2}{\sigma^2+4\,D\,t}\right)}\,D}{\left(1+\frac{4\,D\,t}{\sigma^2}\right)^{(3/2)}\,\sigma^2} + \frac{4\,To\left(x-\frac{1}{2}L\right)^2 D\, \mathbf{e}^{\left(-\frac{(x-1/2\,L)^2}{\sigma^2+4\,D\,t}\right)}}{\sqrt{1+\frac{4\,D\,t}{\sigma^2}}\,(\sigma^2+4\,D\,t)^2}$$

Calculate the spatial derivative on the right

```
> right:=D*diff(T,x$2);
```

$$right := D\left(-2\,\frac{To\, \mathbf{e}^{\left(-\frac{(x-1/2\,L)^2}{\sigma^2+4\,D\,t}\right)}}{\sqrt{1+\frac{4\,D\,t}{\sigma^2}}\,(\sigma^2+4\,D\,t)} + \frac{4\,To\left(x-\frac{1}{2}L\right)^2 \mathbf{e}^{\left(-\frac{(x-1/2\,L)^2}{\sigma^2+4\,D\,t}\right)}}{\sqrt{1+\frac{4\,D\,t}{\sigma^2}}\,(\sigma^2+4\,D\,t)^2}\right)$$

Check to see if the two sides match

```
> simplify(expand(left-right));
```

$$0$$

(b)

```
> restart;
```

Here is the trial form of the solution

```
> T:=sin(n*Pi*x/L)*f(t);
```

$$T := \sin\left(\frac{n\,\pi\,x}{L}\right) f(t)$$

Substitute it into the diffusion equation to get an equation for f(t)

```
> eq:=diff(T,t)=D*diff(T,x$2);
```

$$eq := \sin\left(\frac{n\,\pi\,x}{L}\right)\left(\frac{d}{dt}\,\mathrm{f}(t)\right) = -\frac{\mathrm{D}\,\sin\left(\frac{n\,\pi\,x}{L}\right) n^2\,\pi^2\,\mathrm{f}(t)}{L^2}$$

```
> eq:=simplify(eq/sin(n*Pi*x/L));
```

$$eq := \frac{d}{dt}\,\mathrm{f}(t) = -\frac{\mathrm{D}\,n^2\,\pi^2\,\mathrm{f}(t)}{L^2}$$

Solve for f(t)

```
> dsolve(eq,f(t));
```

$$\mathrm{f}(t) = \_C1\ \mathbf{e}^{\left(-\frac{\mathrm{D}\,n^2\,\pi^2\,t}{L^2}\right)}$$

So it is a decaying exponential with the damping rate given by

```
> gn:=n^2*Pi^2*D/L^2;
```

$$gn := \frac{\mathrm{D}\,n^2\,\pi^2}{L^2}$$

:

(c)

```
> restart;
```

Here is the expression for the width of the distribution

```
> width:=sqrt(sigma^2+4*D*t);
```

$$width := \sqrt{\sigma^2 + 4\,\mathrm{D}\,t}$$

Build the equation that determines the time when the width has doubled

```
> eq:=subs(t=0,width)*2=width;
```

$$eq := 2\sqrt{\sigma^2} = \sqrt{\sigma^2 + 4\,\mathrm{D}\,t}$$

Solve for the doubling time

```
> tdouble:=solve(eq,t);
```

$$tdouble := \frac{3}{4}\,\frac{\sigma^2}{\mathrm{D}}$$

```
>
```