Lab Problem 6.1, 6.2, 6.3(d), 6.3(e), Physics 430

```
> restart;
```

(6.1) Use variables named y(j,n) to make the answer come out close to what we need to code the algorithm

Second-order time derivative

```
> dyt2:=(y(j,n-1)-2*y(j,n)+y(j,n+1))/tau^2;
```

$$dyt2 := \frac{y(j, n-1) - 2\, y(j, n) + y(j, n+1)}{\tau^2}$$

Second-order space derivative

```
> dyx2:=(y(j-1,n)-2*y(j,n)+y(j+1,n))/h^2;
```

$$dyx2 := \frac{y(j-1, n) - 2\, y(j, n) + y(j+1, n)}{h^2}$$

Here is the wave equation tranlated into finite-difference form

```
> wave:=dyt2-c^2*dyx2=0;
```

$$wave := \frac{y(j, n-1) - 2\, y(j, n) + y(j, n+1)}{\tau^2} - \frac{c^2\, (y(j-1, n) - 2\, y(j, n) + y(j+1, n))}{h^2} = 0$$

Now solve for y(j,n+1), the value of y at each grid point one time step into the future

```
> expand(solve(wave,y(j,n+1)));
```

$$-y(j, n-1) + 2\, y(j, n) + \frac{c^2\, \tau^2\, y(j-1, n)}{h^2} - \frac{2\, c^2\, \tau^2\, y(j, n)}{h^2} + \frac{c^2\, \tau^2\, y(j+1, n)}{h^2}$$

(6.2) Write down the two equations and and solve for y(j) one time step into the past

```
> restart;
> eq1:=(y(j,1)-y(j,-1))/2/tau=vj;
```

$$eq1 := \frac{1}{2}\, \frac{y(j, 1) - y(j, -1)}{\tau} = vj$$

```
> eq2:=y(j,1)=2*y(j,0)-y(j,-1)+c^2*tau^2/h^2*(y(j+1,0)-2*y(j,0)+y(j-
  1,0));
```

$$eq2 := y(j, 1) = 2\, y(j, 0) - y(j, -1) + \frac{c^2\, \tau^2\, (y(j+1, 0) - 2\, y(j, 0) + y(j-1, 0))}{h^2}$$

Solve for both y(j,-1) and y(j,1)

```
> solve({eq1,eq2},{y(j,1),y(j,-1)});
```

$$\{ y(j, -1) = -\frac{1}{2}\, \frac{-2\, y(j, 0)\, h^2 + 2\, h^2\, vj\, \tau - c^2\, \tau^2\, y(j+1, 0) + 2\, c^2\, \tau^2\, y(j, 0) - c^2\, \tau^2\, y(j-1, 0)}{h^2},$$

$$y(j, 1) = -\frac{1}{2}\, \frac{-2\, y(j, 0)\, h^2 - 2\, h^2\, vj\, \tau - c^2\, \tau^2\, y(j+1, 0) + 2\, c^2\, \tau^2\, y(j, 0) - c^2\, \tau^2\, y(j-1, 0)}{h^2} \}$$

```
> assign(%);
```
Display the final form
```
> y(j,-1);
```
$$-\frac{1}{2}\frac{-2\,y(j,0)\,h^2+2\,h^2\,vj\,\tau-c^2\,\tau^2\,y(j+1,0)+2\,c^2\,\tau^2\,y(j,0)-c^2\,\tau^2\,y(j-1,0)}{h^2}$$
```
> expand(y(j,-1));
```
$$y(j,0)-vj\,\tau+\frac{\frac{1}{2}c^2\,\tau^2\,y(j+1,0)}{h^2}-\frac{c^2\,\tau^2\,y(j,0)}{h^2}+\frac{\frac{1}{2}c^2\,\tau^2\,y(j-1,0)}{h^2}$$
6.3(d) Rederive staggered leapfrog with damping included

```
> restart;
```
Here is the damping term
```
> damp:=gamma*(y(j,n+1)-y(j,n-1))/(2*tau);
```
$$damp:=\frac{1}{2}\frac{\gamma\,(y(j,n+1)-y(j,n-1))}{\tau}$$
Second-order time derivative
```
> dyt2:=(y(j,n-1)-2*y(j,n)+y(j,n+1))/tau^2;
```
$$dyt2:=\frac{y(j,n-1)-2\,y(j,n)+y(j,n+1)}{\tau^2}$$
Second-order space derivative
```
> dyx2:=(y(j-1,n)-2*y(j,n)+y(j+1,n))/h^2;
```
$$dyx2:=\frac{y(j-1,n)-2\,y(j,n)+y(j+1,n)}{h^2}$$
```
> wave2:=dyt2+damp-c^2*dyx2=0;
```
$$wave2:=\frac{y(j,n-1)-2\,y(j,n)+y(j,n+1)}{\tau^2}+\frac{\frac{1}{2}\gamma\,(y(j,n+1)-y(j,n-1))}{\tau}-\frac{c^2\,(y(j-1,n)-2\,y(j,n)+y(j+1,n))}{h^2}=0$$
```
> y(j,n+1):=solve(wave2,y(j,n+1));
```
$$y(j,n+1):=-(2\,h^2\,y(j,n-1)-4\,h^2\,y(j,n)-\gamma\,\tau\,h^2\,y(j,n-1)-2\,c^2\,\tau^2\,y(j-1,n)+4\,c^2\,\tau^2\,y(j,n)-2\,c^2\,\tau^2\,y(j+1,n))\,/\,(h^2\,(2+\gamma\,\tau))$$
```
> expand(y(j,n+1));
```
$$-2\frac{y(j,n-1)}{2+\gamma\,\tau}+\frac{4\,y(j,n)}{2+\gamma\,\tau}+\frac{\gamma\,\tau\,y(j,n-1)}{2+\gamma\,\tau}+\frac{2\,c^2\,\tau^2\,y(j-1,n)}{h^2\,(2+\gamma\,\tau)}-\frac{4\,c^2\,\tau^2\,y(j,n)}{h^2\,(2+\gamma\,\tau)}$$

$$+\frac{2 c^2 \tau^2 y(j+1, n)}{h^2 (2+\gamma\tau)}$$

This is the damping-modified algorithm. Now find the new starting value of yold

```
> restart;
```

First comes the velocity initial condition

```
> eq1:=(y(j,1)-y(j,-1))/2/tau=vj;
```

$$eq1 := \frac{1}{2}\,\frac{y(j, 1) - y(j, -1)}{\tau} = vj$$

Then we have the damped leapfrog algorithm

```
> eq2:=y(j,1)=-2/(2+gamma*tau)*y(j,-1)+4/(2+gamma*tau)*y(j,0)+1/(2+gamma*tau)*gamma*tau*y(j,-1)+2/h^2/(2+gamma*tau)*c^2*tau^2*y(j-1,0)-4/h^2/(2+gamma*tau)*c^2*tau^2*y(j,0)+2/h^2/(2+gamma*tau)*c^2*tau^2*y(j+1,0);
```

$$eq2 := y(j, 1) = -2\,\frac{y(j, -1)}{2+\gamma\tau} + \frac{4\,y(j, 0)}{2+\gamma\tau} + \frac{\gamma\,\tau\,y(j, -1)}{2+\gamma\tau} + \frac{2 c^2 \tau^2 y(j-1, 0)}{h^2 (2+\gamma\tau)} - \frac{4 c^2 \tau^2 y(j, 0)}{h^2 (2+\gamma\tau)} + \frac{2 c^2 \tau^2 y(j+1, 0)}{h^2 (2+\gamma\tau)}$$

```
> solve({eq1,eq2},{y(j,1),y(j,-1)});
```

$$\{ y(j, 1) = -\frac{1}{2}\,\frac{-2 h^2 vj\,\tau + h^2 \gamma \tau^2 vj - 2\,y(j, 0)\,h^2 - c^2 \tau^2 y(j-1, 0) + 2 c^2 \tau^2 y(j, 0) - c^2 \tau^2 y(j+1, 0)}{h^2},$$

$$y(j, -1) = -\frac{1}{2}\,\frac{2 h^2 vj\,\tau + h^2 \gamma \tau^2 vj - 2\,y(j, 0)\,h^2 - c^2 \tau^2 y(j-1, 0) + 2 c^2 \tau^2 y(j, 0) - c^2 \tau^2 y(j+1, 0)}{h^2} \}$$

```
> assign(%);
```

Here is the final form for y(j,-1)

```
> expand(y(j,-1));
```

$$-vj\,\tau - \frac{1}{2}\gamma\,\tau^2 vj + y(j, 0) + \frac{\frac{1}{2} c^2 \tau^2 y(j-1, 0)}{h^2} - \frac{c^2 \tau^2 y(j, 0)}{h^2} + \frac{\frac{1}{2} c^2 \tau^2 y(j+1, 0)}{h^2}$$

6.3(e) Rederive damped staggered leapfrog with a driving force. Recall that we found the wave-equation form containing $\frac{\partial^2}{\partial t^2} y$ by dividing by the mass density μ, so the driving force must be divided by μ as well.

```
> restart;
```

Here is the damping term

```
> damp:=gamma*(y(j,n+1)-y(j,n-1))/(2*tau);
```

$$damp := \frac{1}{2}\,\frac{\gamma\,(y(j, n+1) - y(j, n-1))}{\tau}$$

Second-order time derivative

```
> dyt2:=(y(j,n-1)-2*y(j,n)+y(j,n+1))/tau^2;
```

$$dyt2 := \frac{y(j, n-1) - 2\,y(j, n) + y(j, n+1)}{\tau^2}$$

Second-order space derivative

```
> dyx2:=(y(j-1,n)-2*y(j,n)+y(j+1,n))/h^2;
```

$$dyx2 := \frac{y(j-1, n) - 2\,y(j, n) + y(j+1, n)}{h^2}$$

```
> wave3:=dyt2+damp-c^2*dyx2=f/mu;
```

$$wave3 := \frac{y(j, n-1) - 2\,y(j, n) + y(j, n+1)}{\tau^2} + \frac{\frac{1}{2}\gamma\,(y(j, n+1) - y(j, n-1))}{\tau} - \frac{c^2\,(y(j-1, n) - 2\,y(j, n) + y(j+1, n))}{h^2} = \frac{f}{\mu}$$

```
> y(j,n+1):=solve(wave3,y(j,n+1));
```

$$y(j, n+1) := (-2\,h^2\,\mu\,y(j, n-1) + 4\,h^2\,\mu\,y(j, n) + \gamma\,\tau\,h^2\,\mu\,y(j, n-1) + 2\,c^2\,\tau^2\,\mu\,y(j-1, n) - 4\,c^2\,\tau^2\,\mu\,y(j, n) + 2\,c^2\,\tau^2\,\mu\,y(j+1, n) + 2\,f\,\tau^2\,h^2) \Big/ (h^2\,\mu\,(2 + \gamma\,\tau))$$

```
> expand(y(j,n+1));
```

$$-2\,\frac{y(j, n-1)}{2 + \gamma\,\tau} + \frac{4\,y(j, n)}{2 + \gamma\,\tau} + \frac{\gamma\,\tau\,y(j, n-1)}{2 + \gamma\,\tau} + \frac{2\,c^2\,\tau^2\,y(j-1, n)}{h^2\,(2 + \gamma\,\tau)} - \frac{4\,c^2\,\tau^2\,y(j, n)}{h^2\,(2 + \gamma\,\tau)} + \frac{2\,c^2\,\tau^2\,y(j+1, n)}{h^2\,(2 + \gamma\,\tau)} + \frac{2\,f\,\tau^2}{\mu\,(2 + \gamma\,\tau)}$$

The term at the end is the new piece due to the driving force. Now find the new starting value of yold

```
> restart;
```

First comes the velocity initial condition

```
> eq1:=(y(j,1)-y(j,-1))/2/tau=vj;
```

$$eq1 := \frac{1}{2}\,\frac{y(j, 1) - y(j, -1)}{\tau} = vj$$

Then we have the damped leapfrog algorithm with the force term added at the back end

```
> eq2:=y(j,1)=-2/(2+gamma*tau)*y(j,-1)+4/(2+gamma*tau)*y(j,0)+1/(2+g
  amma*tau)*gamma*tau*y(j,-1)+2/h^2/(2+gamma*tau)*c^2*tau^2*y(j-1,0)
  -4/h^2/(2+gamma*tau)*c^2*tau^2*y(j,0)+2/h^2/(2+gamma*tau)*c^2*tau^
  2*y(j+1,0)+2/(2+gamma*tau)*f*tau^2/mu;
```

$$eq2 := y(j, 1) = -2\,\frac{y(j, -1)}{2 + \gamma\,\tau} + \frac{4\,y(j, 0)}{2 + \gamma\,\tau} + \frac{\gamma\,\tau\,y(j, -1)}{2 + \gamma\,\tau} + \frac{2\,c^2\,\tau^2\,y(j-1, 0)}{h^2\,(2 + \gamma\,\tau)} - \frac{4\,c^2\,\tau^2\,y(j, 0)}{h^2\,(2 + \gamma\,\tau)} + \frac{2\,c^2\,\tau^2\,y(j+1, 0)}{h^2\,(2 + \gamma\,\tau)} + \frac{2\,f\,\tau^2}{(2 + \gamma\,\tau)\,\mu}$$

```
> solve({eq1,eq2},{y(j,1),y(j,-1)});
```

$$\{ y(j,1) = -\frac{1}{2}(-2\,h^2\,\mu\,vj\,\tau + h^2\,\mu\,\gamma\,\tau^2\,vj - 2\,y(j,0)\,h^2\,\mu - c^2\,\tau^2\,y(j-1,0)\,\mu + 2\,c^2\,\tau^2\,y(j,0)\,\mu$$

$$- c^2\,\tau^2\,y(j+1,0)\,\mu - f\,\tau^2\,h^2) \Big/ (h^2\,\mu),\ y(j,-1) = -\frac{1}{2}(2\,h^2\,\mu\,vj\,\tau + h^2\,\mu\,\gamma\,\tau^2\,vj - 2\,y(j,0)\,h^2\,\mu$$

$$- c^2\,\tau^2\,y(j-1,0)\,\mu + 2\,c^2\,\tau^2\,y(j,0)\,\mu - c^2\,\tau^2\,y(j+1,0)\,\mu - f\,\tau^2\,h^2) \Big/ (h^2\,\mu) \}$$

```
> assign(%);
```

Here is the final form for y(j,-1)

```
> y(j,-1):=expand(y(j,-1));
```

$$y(j,-1) :=$$

$$-vj\,\tau - \frac{1}{2}\gamma\,\tau^2\,vj + y(j,0) + \frac{\frac{1}{2}c^2\,\tau^2\,y(j-1,0)}{h^2} - \frac{c^2\,\tau^2\,y(j,0)}{h^2} + \frac{\frac{1}{2}c^2\,\tau^2\,y(j+1,0)}{h^2} + \frac{\frac{1}{2}f\,\tau^2}{\mu}$$

$$-vj\,\tau - \frac{1}{2}\gamma\,\tau^2\,vj + y(j,0) + \frac{\frac{1}{2}c^2\,\tau^2\,y(j-1,0)}{h^2} - \frac{c^2\,\tau^2\,y(j,0)}{h^2} + \frac{\frac{1}{2}c^2\,\tau^2\,y(j+1,0)}{h^2}$$

But in the problem at hand all of the y(j)'s as well as the initial velocity are zero. So in this simple case y(j,-1) reduces to

```
> subs(vj=0,y(j,0)=0,y(j+1,0)=0,y(j-1,0)=0,y(j,-1));
```

$$\frac{1}{2}\frac{f\,\tau^2}{\mu}$$

which is just our old friend from freshman physics, $\frac{1\,a\,t^2}{2}$.

```
>
>
```