Lab Problem 2.4, Physics 430

```
> restart;
```

(a)

Make expressions for f(x+p), f(x), and f(x-m) with variables in them that will allow us to solve for f' and f''

```
> eqplus:=fplus=f + p*fp + 1/2*p^2*fpp+1/6*p^3*fppp+1/24*p^4*fpppp;
```

$$eqplus := fplus = f + p\,fp + \frac{1}{2}p^2\,fpp + \frac{1}{6}p^3\,fppp + \frac{1}{24}p^4\,fpppp$$

```
> eq0:=f0=f;
```

$$eq0 := f0 = f$$

```
> eqminus:=fminus=f - m*fp +
  1/2*m^2*fpp-1/6*m^3*fppp+1/24*m^4*fpppp;
```

$$eqminus := fminus = f - m\,fp + \frac{1}{2}m^2\,fpp - \frac{1}{6}m^3\,fppp + \frac{1}{24}m^4\,fpppp$$

(b)

```
> sol:=solve({eqplus,eq0,eqminus},{f,fp,fpp});
```

$$sol := \{f = f0,\ fpp = \frac{1}{12}(24\,p\,fminus - 24\,p\,f0 - 24\,m\,f0 + 4\,p\,m^3\,fppp - p\,m^4\,fpppp + 24\,m\,fplus - 4\,m\,p^3\,fppp - m\,p^4\,fpppp)\,/\,(p\,m\,(m+p)),\ fp = \frac{1}{24}(24\,m^2\,fplus - 24\,m^2\,f0 - 24\,p^2\,fminus + 24\,p^2\,f0 - 4\,p^2\,m^3\,fppp + p^2\,m^4\,fpppp - 4\,m^2\,p^3\,fppp - m^2\,p^4\,fpppp)\,/\,(p\,m\,(m+p))\}$$

```
> assign(sol);
```

Now look at each one and see how big it's error is by assuming that the higher order derivatives are unknown (set them to zero)

f=f0 is obvious

```
> f;
```

$$f0$$

(c)

Here's the formula for fp:

```
> fpapprox:=subs(fppp=0,fpppp=0,fp);
```

$$fpapprox := \frac{1}{24}\,\frac{24\,m^2\,fplus - 24\,m^2\,f0 - 24\,p^2\,fminus + 24\,p^2\,f0}{p\,m\,(m+p)}$$

Now find the error by subtracting this approximation from the form with higher derivatives in it

```
> err:=simplify(fp-fpapprox);
```

$$err := -\frac{1}{24}(fpppp\,p + 4\,fppp - fpppp\,m)\,p\,m$$

So the error is of order p*m, second order, and if p=m the error is

```
> erreq:=subs(m=p,err);
```

$$erreq := -\frac{1}{6}\, fppp\, p^2$$

still just second order in the stepsize. And here is the derivative formula with p=m

```
> fpeq:=simplify(subs(m=p,fpapprox));
```

$$fpeq := \frac{1}{2}\,\frac{fplus - fminus}{p}$$

Now do the same thing for the second derivative.

```
> fppapprox:=subs(fppp=0,fpppp=0,fpp);
```

$$fppapprox := \frac{1}{12}\,\frac{24\,p\,fminus - 24\,p\,f0 - 24\,m\,f0 + 24\,m\,fplus}{p\,m\,(m+p)}$$

```
> err:=simplify(simplify(fpp-fppapprox));
```

$$err := -\frac{1}{12}\,p^2\, fpppp - \frac{1}{3}\,p\, fppp + \frac{1}{12}\,p\, fpppp\, m + \frac{1}{3}\,m\, fppp - \frac{1}{12}\,m^2\, fpppp$$

This error is first order in p and m, unless they are equal, in which case we have

```
> fppeq:=simplify(subs(m=p,fppapprox));erreq:=simplify(subs(m=p,err)
  );
```

$$fppeq := \frac{fminus - 2\,f0 + fplus}{p^2}$$

$$erreq := -\frac{1}{12}\,p^2\, fpppp$$

So only with equal spacing do we get a second-order error term. This is important and we will use equally spaced grids throughout this course.

(d) Write down the Taylor expansion at each of the 4 points, then solve for [f,fp,fpp,fppp]

```
> restart;
> d:=h/2;
```

$$d := \frac{1}{2}\,h$$

```
> eqp:=fup=f+fp*d+fpp*d^2/2+fppp*d^3/6+fpppp*d^4/24+fp5*d^5/120;
```

$$eqp := fup = f + \frac{1}{2}\,fp\,h + \frac{1}{8}\,fpp\,h^2 + \frac{1}{48}\,fppp\,h^3 + \frac{1}{384}\,fpppp\,h^4 + \frac{1}{3840}\,fp5\,h^5$$

```
> d:=3*h/2;
```

$$d := \frac{3}{2}\,h$$

```
> eqpp:=fupper=f+fp*d+fpp*d^2/2+fppp*d^3/6+fpppp*d^4/24+fp5*d^5/120;
```

$$eqpp := fupper = f + \frac{3}{2}\,fp\,h + \frac{9}{8}\,fpp\,h^2 + \frac{9}{16}\,fppp\,h^3 + \frac{27}{128}\,fpppp\,h^4 + \frac{81}{1280}\,fp5\,h^5$$

$$eqp := fp = f + fp\, d + \frac{1}{2}\, fpp\, d^2 + \frac{1}{6}\, fppp\, d^3 + \frac{1}{24}\, fpppp\, d^4 + \frac{1}{120}\, fp5\, d^5$$

```
> d:=-h/2;
```

$$d := -\frac{1}{2}\, h$$

```
> eqm:=fdown=f+fp*d+fpp*d^2/2+fppp*d^3/6+fpppp*d^4/24+fp5*d^5/120;
```

$$eqm := fdown = f - \frac{1}{2}\, fp\, h + \frac{1}{8}\, fpp\, h^2 - \frac{1}{48}\, fppp\, h^3 + \frac{1}{384}\, fpppp\, h^4 - \frac{1}{3840}\, fp5\, h^5$$

$$eqp := fp = f + fp\, d + \frac{1}{2}\, fpp\, d^2 + \frac{1}{6}\, fppp\, d^3 + \frac{1}{24}\, fpppp\, d^4 + \frac{1}{120}\, fp5\, d^5$$

```
> d:=-3*h/2;
```

$$d := -\frac{3}{2}\, h$$

```
> eqmm:=fdowner=f+fp*d+fpp*d^2/2+fppp*d^3/6+fpppp*d^4/24+fp5*d^5/120
;
```

$$eqmm := fdowner = f - \frac{3}{2}\, fp\, h + \frac{9}{8}\, fpp\, h^2 - \frac{9}{16}\, fppp\, h^3 + \frac{27}{128}\, fpppp\, h^4 - \frac{81}{1280}\, fp5\, h^5$$

$$eqp := fp = f + fp\, d + \frac{1}{2}\, fpp\, d^2 + \frac{1}{6}\, fppp\, d^3 + \frac{1}{24}\, fpppp\, d^4 + \frac{1}{120}\, fp5\, d^5$$

```
> s:=solve({eqp,eqpp,eqm,eqmm},{f,fp,fpp,fppp});
```

$$s := \left\{ fppp = -\frac{1}{8}\, \frac{8\, fdowner - 8\, fupper + fp5\, h^5 + 24\, fup - 24\, fdown}{h^3}, \right.$$

$$fpp = -\frac{1}{24}\, \frac{-12\, fupper + 12\, fup - 12\, fdowner + 12\, fdown + 5\, fpppp\, h^4}{h^2},$$

$$fp = \frac{1}{1920}\, \frac{-2160\, fdown + 80\, fdowner - 80\, fupper + 9\, fp5\, h^5 + 2160\, fup}{h},$$

$$\left. f = \frac{9}{16}\, fup + \frac{3}{128}\, fpppp\, h^4 + \frac{9}{16}\, fdown - \frac{1}{16}\, fdowner - \frac{1}{16}\, fupper \right\}$$

```
> assign(s);
> expand(fppp);
```

$$-\frac{fdowner}{h^3} + \frac{fupper}{h^3} - \frac{1}{8}\, h^2\, fp5 - \frac{3\, fup}{h^3} + \frac{3\, fdown}{h^3}$$

This is the 1, -3, 3, 1 rule and shows that it is second order accurate

```
>
```